
ANODE_LOOK_ELC_DEFL_EQNS_V01 
 
This document provides equations to calculate the High Rate Science (HRS) 
elevations angles for JADE-E when the deflectors are utilized : version 01.  This is a 
simplified version of what the onboard software does.  Later versions may improve 
this to be closer to the onboard software. 
 
The first part is a description of the process, the second part is the IDL code used to 
represent the process (as used in the JADE product code for this version), albeit it 
with some cosmetic differences to look better on a page and not wrap lines, e.g. 
in_file.MAG_VECTOR[*,0]^2 instead of 

in_file.MAG_VECTOR[*,0]*in_file.MAG_VECTOR[*,0]). 
 
The electron deflection equations only affect elevation angles of the field-of-view 
(FOV).  For azimuthal angles (with or without deflection) see the relevant version of 
file ANODE_LOOK_ELC_DEFL_NONE_Vnn.  To know which version (for either 
elevation or azimuth) was used for a given Level 3 JADE record, look at the 
SOURCE_JADE_CALIB object which is a version number, mmmmm, which 
corresponds to the file JAD_L30_CALIB_LIST_mmmmm.TXT.  This file then says 
which versions of other files (such as file ANODE_LOOK_ELC_DEFL_NONE_Vnn or 
ANODE_LOOK_ELC_DEFL_EQNS_Vnn) were used. 
 
Note: Electron deflectors for HRS were not used for science data during Cruise, nor 
JOI, their first science use was at Perijove 1 (although they had previously been used 
during operations check-out tests, but those are not for science use).  Any HRS 
before that time (or if the JADE Level 2 data record has MAG_VECTOR = [0, 0, 0]) did 
not deflect, and the non-deflection elevation angles should be used, which are found 
in the relevant version of file ANODE_LOOK_ELC_DEFL_NONE_Vnn (see 
JAD_L30_CALIB_LIST_mmmmm.TXT for which nn). 

Deflection Description 
The following description is mostly a cut & paste from internal JADE document 
“from_B_to_elevation_20161020.doc”.  The “Broadcast magnetic field” is in 
spacecraft co-ordinates, and is uncalibrated.  It does not list the FWHM of Elevation 
angle for a given deflection, which is given in separate internal JADE documentation 
(“JADE-E_calibration_report_v1.docx”) and is based on ground calibration data and 
electro-optics simulations; relevant parts are also cut & pasted here. 

 
JADE-E elevation of look direction in HRS 

Frédéric Allegrini, Sept. 1, 2016 
 
At the beginning of every energy sweep, JADE uses the propagated broadcast 
magnetic field to determine the elevation of the look direction and which sensor 
uses its deflectors: 
 
Broadcast magnetic field: B = (Bx, By, Bz) (1) 

 
 
Step 0: determine the sign of the deflection angle (positive or negative) for each 
sensor. 
 
At the beginning of the sweep (every second), the propagated broadcast magnetic 
field falls onto an anode (called the “mag_anode”, integer between 0 and 47): if J is 
the structure containing the HRS data, then  
 
mag_anode = J.MAG_LOOK_DIR (2) 

Step 1: calculate elevation, , in degrees of B in S/C coordinates 

B is capped at 35°.  
If B > 35°, then B =35° 
If B < -35°, then B =-35° 
 
To determine the targeted elevation (i.e., the angle that JADE-E is deflecting to), D, 
for this second we use the following:  

 For E060: if 20≤ mag_anode≤43, then D = -B, else D = B 
 For E180: if 12≤ mag_anode≤35, then D = B, else D = -B 

 
D is the angle to use for now in the conversion of L2 to L3 JADE-E data. It is the 
same for all anodes and energies for a given second. 
 
This shortened elevation calculation does not account for offsets from the different 
conversions within the instrument, errors due to HVPS not reaching targeted 
voltages, and magnetic field effects on the trajectories. 
 
The steps from the previous version of this document (which took some of the 
offsets from conversions into account) and refinements to the method will be 
implemented later as needed.  
 
 
 

The FWHM of Elevation angle is explained on the next page. 

Equation 15 of “JADE-E_calibration_report_v1.docx” provides the elevation (el) 
resolution equations as shown below (units of degrees), and their Figure 24 (used 
as the source of the equations) is pasted below that: 
 

 
 
 

 
Figure 24. FWHM of the elevation angle distribution as a function of elevation angle 
from simulations at energies from 102 to 105 eV. The thick lines are fits using Eq. 
15. 
 
 
These equations are used in IDL for the DIM2_ELEVATION_UPPER and 
DIM2_ELEVATION_UPPER objects of JADE Level 3 files. 

IDL code used to populate JADE Level 3 files 
The code below was used to calculation the center elevation angles 
(DIM2_ELEVATION) only if there was deflection. 
 
In_file is a structure of the objects, which includes fields of: 

 MAG_VECTOR : 
A vector (size 3) of the magnetic field in spacecraft co-ordinates (JADE 
despun spacecraft or regular spacecraft, does not matter, as identical +z axis) 

 MAG_LOOK_DIR: 
Mag_look_dir = MAG Look direction (0 to 47) from JADE-E  HRS Level 2 files 

Specific IDL coding methods: 
 IDL uses $ to continue on next line. 
 Doubles are denoted with a d, e.g. 35d = double(35).  
 Signed long integers (4-bytes) are denoted with an L (e.g. -1L = long(-1)). 
 Numbers without a d or an L are signed short integers (2-bytes). 
 EQ = equal, NE = not equal. 
 GT = greater than, GE = greater than or equal. 
 LT = less than, LE = less than or equal. 

 
; If HRS_ALL data was used, the E300 ones will be 65535 so not altered. 

Elevation_delta = _hrs_electron_deflection_simple(L3,L2.MAG_LOOK_DIR) 

; returns -1 if too early for deflection, i.e. cruise 

 

; Preallocate the three elevation arrays with fill values of double(65535) 

L3.DIM2_ELEVATION = Elevation_delta ; just to get array of correct size 

L3.DIM2_ELEVATION[*] = 65535d ; set all to fill 

L3.DIM2_ELEVATION_UPPER = L3.DIM2_ELEVATION ; copy 

L3.DIM2_ELEVATION_LOWER = L3.DIM2_ELEVATION ; copy 

 

IF Elevation_delta[0] NE -1 THEN BEGIN 

  ind = WHERE(Elevation_delta LT 65534d,/NULL) 

  ; using 65534 to avoid rounding issues, expect values < 35, so very safe 

   

  IF N_ELEMENTS(ind) GT 0 THEN BEGIN 

    ; First do center values: 

    L3.DIM2_ELEVATION[ind] = Elevation_delta[ind] 

 

    ; Work out Delta's, first assume all is in the middle range, 

    ; then do the two end ranges, if any. 

    Elevation_delta_FWHM = 3.55d - 0.0606d * Elevation_delta 

     

    ind1 = WHERE( Elevation_delta LT -32.2d,/NULL) 

    IF (N_ELEMENTS(ind1) GT 0) THEN $ 

        Elevation_delta_FWHM[ind1] =   29.9d + 0.758d * Elevation_delta[ind1] 

 

    ind1 = WHERE( (Elevation_delta GE 23.4d) AND (Elevation_delta LT 65534d), /NULL) 

                                                ; need to ignore fill values 

    IF (N_ELEMENTS(ind1) GT 0) THEN $ 

        Elevation_delta_FWHM[ind1] = -0.189d +0.0993d * Elevation_delta[ind1] 

 

    ; Divide by 2 to get half FWHM for upper and lower 

    Elevation_delta_FWHM = Elevation_delta_FWHM / 2d 

    ; base the delta's off the new center energy 

    L3.DIM2_ELEVATION_UPPER[ind] = L3.DIM2_ELEVATION[ind] + Elevation_delta_FWHM[ind] 

    L3.DIM2_ELEVATION_LOWER[ind] = L3.DIM2_ELEVATION[ind] - Elevation_delta_FWHM[ind] 

  ENDIF 

ENDIF 

 

The function _hrs_electron_deflection_simple is given on the next page.  

FUNCTION _hrs_electron_deflection_simple, in_file, Mag_look_dir 

 

  ; IDL compilation bits – not important 

  COMPILE_OPT HIDDEN 

  ON_ERROR,2   ; Quit if MESSAGE runs 

 

  ; Check input structure in_file is that of HRS data (PACKETID = 142) 

  IF in_file.PACKETID[0] NE 142 THEN $  

    MESSAGE,'ERROR: _hrs_electron_deflection only applies to HRS Electron datasets' 

 

; IF DATE BEFORE JOI THEN RETURN, DEFLECTION WAS NEVER ON 

; First HRS electron data with deflectors was 2016-240 (2016-Aug-27) 

  IF in_file.TIMESTAMP_WHOLE[0] LT 520862570 THEN BEGIN ; 520862570 = 2016-Jul-04, JOI 

    PRINT,'Time stamp before JOI, was no HRS deflection to be done.  Returning as is.' 

    RETURN,-1 ; Use non-deflection Electron values. 

  ENDIF 

 

  ; IF AFTER JOI, ONLY E060 and E180 where on, never E300 

 

  ; pre-allocate array of fill values (65535) for elevation delta 

  Elevation_delta = in_file.DIM2_ELEVATION ; get right size, be it 32 or 48 anodes 

  Elevation_delta[*] = 65535d ; make all fill, 65535 as a double (the d) 

 

  ; Step 0  ; mag_anode must be 0-47 only - enforced before generation of Level 2 files! 

  mag_anode = ROUND( Mag_look_dir ) ; Round to integer (avoid rounding errors on doubles) 

 

  ; Step 1, find elevation in degrees and cap at +/- 35 degs 

  rads2degs = 180d/!DPI ; = 180/pi 

  deltaB = rads2degs * ATAN(in_file.MAG_VECTOR[*,2] / $ 

           SQRT(in_file.MAG_VECTOR[*,0]^2 +in_file.MAG_VECTOR[*,1]^2 ) ) ; in degrees 

  deltaB( WHERE(deltaB GT  35d ,/NULL) ) =  35d ; upper cap 

  deltaB( WHERE(deltaB LT -35d ,/NULL) ) = -35d ; lower cap 

 

  ; Step 2 - assign elevation based on anode and mag_look_dir 

  ; Do For loop through array as MAG vector can change 

  FOR rec = 0L,(N_ELEMENTS(in_file.T) - 1L)  DO BEGIN 

    ; ignore if MAG vector is all zeros or fills (leave Elevation_delta as 65535) 

    ; Check for all zeros 

    IF ((in_file.MAG_VECTOR[rec,0] EQ 0L) AND (in_file.MAG_VECTOR[rec,1] EQ 0L) AND $ 

        (in_file.MAG_VECTOR[rec,2] EQ 0L)) THEN CONTINUE; skip if no MAG vector ([0,0,0]) 

    ; Then check for any MAG_VECTOR fill values, 2147483647, but to avoid rounding errors 

    ; use 1 less and GT (>), so 2147483647-1 = 2147483646 

    IF (in_file.MAG_VECTOR[rec,0] GT 2147483646L) THEN CONTINUE 

    IF (in_file.MAG_VECTOR[rec,1] GT 2147483646L) THEN CONTINUE 

    IF (in_file.MAG_VECTOR[rec,2] GT 2147483646L) THEN CONTINUE 

 

    ; Still in For Loop – so MAG vector is not zeros nor fill values. 

 

    ; Now do elevation for E060 

    IF ((mag_anode[rec] GE 20) AND (mag_anode[rec] LE 43)) THEN BEGIN 

         Elevation_delta[rec,*, 0:15] = -deltaB[rec] 

    ENDIF ELSE BEGIN 

         Elevation_delta[rec,*, 0:15] =  deltaB[rec]  

    ENDELSE 

    ; Now do elevation for E180 

    IF ((mag_anode[rec] GE 12) AND (mag_anode[rec] LE 35)) THEN BEGIN 

         Elevation_delta[rec,*,16:31] =  deltaB[rec] 

    ENDIF ELSE BEGIN 

         Elevation_delta[rec,*,16:31] = -deltaB[rec 

    ENDELSE 

    ; Nothing to do for E300, it's all fills – which is already pre-allocated 

  ENDFOR 

 

  RETURN,Elevation_delta 

END 

